§7 Computerunterstütztes Abzählen

▼

```
PermutationenOhneWiederholung[n_Integer] :=
  Permutations[Range[n]]

PermutationenMitWiederholung[n_List] :=
  Permutations[Flatten[Table[Table[i, {n[[i]]}], {i, Length[n]}]]]

KombinationenOhneWiederholung[n_Integer, k_Integer] :=
  Permutations[Join[Table[1, {k}], Table[0, {n – k}]]]

KombinationenOhneWiederholungAlt[n_Integer, 1] := Table[{i}, {i, 1, n}]
KombinationenOhneWiederholungAlt[n_Integer, k_Integer] :=
  Flatten[Map[Table[Append[#, i], {i, Last[#] + 1, n}] &, KombinationenOhneWiederholungAlt[n, k – 1]], 1]

KombinationenMitWiederholung[1, k_Integer] := {{k}}
KombinationenMitWiederholung[n_Integer, k_Integer] :=
  Flatten[Table[Map[Append[#, i] &, KombinationenMitWiederholung[n – 1, k – i]], {i, 0, k}], 1]

KombinationenMitWiederholungAlt[n_Integer, 1] := Table[{i}, {i, 1, n}]
KombinationenMitWiederholungAlt[n_Integer, k_Integer] :=
  Flatten[Map[Table[Append[#, i], {i, Last[#], n}] &, KombinationenMitWiederholungAlt[n, k – 1]], 1]

VariationenOhneWiederholung[n_Integer, k_Integer] :=
  Flatten[Map[Permutations, KombinationenOhneWiederholungAlt[n, k]], 1]

VariationenMitWiederholung[n_Integer, k_Integer] :=
  Distribute[Table[Table[i, {i, 1, n}], {k}], List]
```

Wie wir bereits wissen, läuft die Berechnung der Wahrscheinlichkeit eines Ereingisses $A \subseteq \Omega$ bei Laplace-Experimenten auf die Berechnung der Mächtigkeit der Mengen A und Ω hinaus. In vielen Fällen handelt es sich bei der Menge Ω dabei um eine Menge, welche wir in Kapitel 6 bereits kennen gelernt haben und die wir mit Hilfe von Mathematica erzeugen können.

Oft lässt sich die Menge A dann einfach dadurch erzeugen, indem wir mit den Befehlen Select bzw Cases aus dieser Menge Ω alle jene Elemente auswählen, welche eine bestimmte Bedingung erfüllen bzw ein bestimmtes Muster besitzen. Zur Berechnung der gesuchten Wahrscheinlichkeit $\mathbb{P}[A]$ muss man dann nur mehr mit Length die Mächtigkeiten der beiden Mengen A und Ω ermitteln und von diesen Mächtigkeiten den Quotient berechnen.

7.1 Beispiele

Wir werden nun an Hand einiger der bereits in Kapitel 4 behandelten Beispiele zeigen, wie man die Wahrscheinlichkeit eines Ereignisses $A \subseteq \Omega$ durch computerunterstütztes Abzählen ermitteln kann. Üblicherweise besteht dieser Vorgang in den folgenden drei Schritten:

- Genaue Angabe des Ereignisraums Ω und der das interessierende Ereignis beschreibenden Menge A:

Die genaue Angabe des Ereignisraums Ω des zur Diskussion stehenden Zufallsexperiments erfordert eine detaillierte Analyse dieses Zufallsexperiments. Erst dadurch wird in vielen Fällen klar, welche Realisierungen bei diesem Zufallsexperiment tatsächlich eintreten, wie man diese Realisierungen beschreibt und ob man annehmen kann, dass diese Realisierungen stets mit der gleichen Wahrscheinlichkeit auftreten.

- Erzeugung der Mengen Ω und A mit den Befehlen aus Kapitel 6 sowie den Befehlen Select bzw Cases:

In vielen Fällen handelt es sich bei der Menge Ω um die Menge aller Permutationen bzw Kombinationen bzw

Variationen mit bzw ohne Wiederholung. Wie sich diese Mengen mit Hilfe von *Mathematica* erzeugen lassen, haben wir in Kapitel 6 ausführlich beschrieben. Die Auswahl des interessierenden Ereignisses A mit Hilfe von Select bzw Cases ist dann meist trivial. Da die Mengen Ω und A oft mehrere Tausend Elemente besitzen, ist es nicht sinnvoll, diese Mengen explizit auszugeben.

- Ermittlung der Wahrscheinlichkeit des uns interessierenden Ereignisses:
Die Mächtigkeiten der Mengen A und Ω lassen sich dann leicht mit Hilfe von Length ermitteln. Die gesuchte Wahrscheinlichkeit ergibt sich schließlich als Quotient der Mächtigkeiten dieser beiden Mengen.

Zum besseren Verständnis ist es oft wünschenswert, für sehr kleine Werte von k, m, … die Mengen Ω und A explizit aufzulisten. Man entferne dazu einfach das Semikolon am Ende der Zuweisungen "omega" bzw "ereignis".

7.1.1 Beispiel: $k = 4$ Würfel werden gleichzeitig geworfen. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei die maximale Augenzahl kleiner als $m = 5$ ist? (Vergleiche Beispiel 4.2.1.)

▼

Lösung: Das Zufallsexperiment besteht im gleichzeitigen Werfen von $k = 4$ Würfeln. Dieses Zufallsexperiment besitzt den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{1, 2, \ldots, 6\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "mit dem ersten Würfel wird die Augenzahl x_1, mit dem zweiten Würfel wird die Augenzahl x_2, …, mit dem k-ten Würfel wird die Augenzahl x_k gewürfelt" entspricht. Ein einfaches Gedankenexperiment zeigt, dass alle Realisierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das Ereignis "die maximale Augenzahl ist kleiner als $m = 5$" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid \max[x_1, x_2, \ldots, x_k] < m\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Max[#] < m & auswählen.

```
k = 4; m = 5;
omega = VariationenMitWiederholung[6, k];
ereignis = Select[omega, Max[#] < m &];
N[Length[ereignis] / Length[omega]]
Clear[k, m, omega, ereignis]
```

```
0.197531
```

7.1.2 Beispiel: $k = 6$ Würfel werden gleichzeitig geworfen. Wie groß ist die Wahrscheinlichkeit dafür, dass ihre Augensumme größer als $s = 25$ ist? (Vergleiche Beispiel 4.2.2.)

▼

Lösung: Das Zufallsexperiment besteht im gleichzeitigen Werfen von $k = 6$ Würfeln. Für dieses Zufallsexperiment haben wir bereits (vgl Beispiel 7.1.1) mit

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{1, 2, \ldots, 6\}\}$$

einen passenden Ereignisraum gefunden. Das Ereignis "die Augensumme ist größer als $s = 25$" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid x_1 + x_2 + \ldots + x_k > s\}$$

Die Menge Ω lässt sich wieder mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Apply[Plus, #] > s & auswählen.

```
k = 6; s = 25;
omega = VariationenMitWiederholung[6, k];
ereignis = Select[omega, Apply[Plus, #] > s &];
N[Length[ereignis] / Length[omega]]
Clear[k, s, omega, ereignis]
```

```
0.144633
```

7.1.3 Beispiel: $k = 15$ Münzen werden gleichzeitig geworfen. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei weniger als $a = 7$ Adler auftreten? (Vergleiche Beispiel 4.2.3.)

▼

Lösung: Das Zufallsexperiment besteht im gleichzeitigen Werfen von $k = 15$ Münzen. Dieses Zufallsexperiment besitzt den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{0, 1\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "mit der i-ten Münze wird genau dann ein Adler geworfen, wenn $x_i = 1$ ist" entspricht. Ein einfaches Gedankenexperiment zeigt, dass alle Realisierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das Ereignis "es treten dabei weniger als $a = 7$ Adler auf" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid x_1 + x_2 + \ldots + x_k < a\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Apply[Plus, #] < a & auswählen.

```
k = 15; a = 7;
omega = VariationenMitWiederholung[2, k] − 1;
ereignis = Select[omega, Apply[Plus, #] < a &];
N[Length[ereignis] / Length[omega]]
Clear[k, a, omega, ereignis]
```

```
0.303619
```

7.1.4 Beispiel: Zwei gleich gute Spieler spielen bis zum Gesamtsieg. Dazu muss der erste Spieler noch $a = 8$ und der zweite Spieler noch $b = 6$ Runden gewinnen. Man bestimme die Wahrscheinlichkeit dafür, dass der erste Spieler gewinnt? (Vergleiche Beispiel 4.2.4.)

▼

Lösung: Wir kommen überein, einen Sieg des ersten Spielers mit 1 und einen Sieg des zweiten Spielers mit 0 zu bezeichnen. Eine Entscheidung darüber, welcher Spieler den Gesamtsieg davon trägt, ist nach höchstens $k = a + b - 1$ Runden entschieden. Unser Zufallsexperiment besteht somit in der Beobachtung des Spielverlaufes von k Runden. Dieses Zufallsexperiment besitzt den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{0, 1\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "der erste Spieler gewinnt die i-te Runde genau dann, wenn $x_i = 1$ ist" entspricht. Da die beiden Spieler voraussetzungsgemäß gleich gut spielen, sind alle Realisierungen $\omega \in \Omega$ gleich wahrscheinlich. Das Ereignis "der erste Spieler erzielt den Gesamtsieg" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid \sum_{i=1}^{k} x_i \geq a\}$$

Die Menge Ω lässt sich wieder mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Apply[Plus, #] ≥ a & auswählen.

```
a = 8; b = 6; k = a + b - 1;
omega = VariationenMitWiederholung[2, k] - 1;
ereignis = Select[omega, Apply[Plus, #] ≥ a &];
N[Length[ereignis] / Length[omega]]
Clear[a, b, k, omega, ereignis]
```

```
0.290527
```

7.1.5 Beispiel: Eine Münze wird $k = 15$ man geworfen und die Anzahl der auftretenden Runs bestimmt. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei zwischen $r_1 = 5$ und $r_2 = 8$ Runs auftreten? (Vergleiche Beispiel 4.2.6.)

▼

Lösung: Das Zufallsexperiment besteht im $k = 15$ maligen Werfen einer Münze. Dieses Zufallsexperiment besitzt bekanntlich den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{0, 1\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "beim i-ten Wurf wird genau dann ein Adler geworfen, wenn $x_i = 1$ ist" entspricht. Offenbar sind alle Realisierungen $\omega \in \Omega$ gleich wahrscheinlich. Das Ereignis "es treten dabei $r_1 \leq r \leq r_2$ Runs auf" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid r_1 \leq \text{Anzahl der Runs der Liste } \{x_1, x_2, \ldots, x_k\} \leq r_2\}$$

Die Menge Ω lässt sich wieder mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung $r_1 \leq$ Length[Split[#]] $\leq r_2$ & auswählen.

```
k = 15; r1 = 5; r2 = 8;
omega = VariationenMitWiederholung[2, k] - 1;
ereignis = Select[omega, r1 ≤ Length[Split[#]] ≤ r2 &];
N[Length[ereignis] / Length[omega]]
Clear[k, r1, r2, omega, ereignis]
```

```
0.57605
```

7.1.6 Beispiel: Aus einer Urne mit $m = 8$ Kugeln werden mit Zurücklegen $k = 6$ Kugeln gezogen. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei mehr als $v = 4$ verschiedene Kugeln gezogen werden? (Vergleiche Beispiel 4.2.7.)

▼

Lösung: Das zufällige Ziehen mit Zurücklegen von $k = 6$ Kugeln aus einer Urne mit $m = 8$ Kugeln besitzt den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{1, 2, \ldots, m\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "beim i-ten Zug wird die Kugel mit der Nummer x_i gezogen" entspricht und alle diese Realisierungen $\omega \in \Omega$ offenbar gleich wahrscheinlich sind. Das Ereignis "es werden dabei mehr als $v = 4$ verschiedene Kugeln gezogen" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid \text{Mächtigkeit der Menge } \{x_1, x_2, \ldots, x_k\} \text{ ist größer als } v\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Length[Union[#]] $> v$ & auswählen.

```
k = 6; m = 8; v = 4;
omega = VariationenMitWiederholung[m, k];
ereignis = Select[omega, Length[Union[#]] > v &];
N[Length[ereignis] / Length[omega]]
Clear[k, m, v, omega, ereignis]
```

```
0.461426
```

7.1.7 Beispiel: Aus einer Urne mit $m = 10$ Kugeln werden $k = 5$ Kugeln gezogen, wobei die gezogene Kugel stets wieder in die Urne zurückgelegt wird. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei mindestens eine Kugel mehrfach gezogen wird? (Vergleiche Beispiel 4.2.8.)

▼

Lösung: Für das zufällige Ziehen mit Zurücklegen von $k = 5$ Kugeln aus einer Urne mit $m = 10$ Kugeln ist bekanntlich

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{1, 2, \ldots, m\}\}$$

ein passender Ereignisraum, wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "beim i-ten Zug wird die Kugel mit der Nummer x_i gezogen" entspricht und alle diese Realisierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das Ereignis "es wird dabei mindestens eine Kugel mehrfach gezogen" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid \text{Mächtigkeit der Menge } \{x_1, x_2, \ldots, x_k\} \text{ ist kleiner als } k\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Length[Union[#]] < k & auswählen.

```
k = 5; m = 10;
omega = VariationenMitWiederholung[m, k];
ereignis = Select[omega, Length[Union[#]] < k &];
N[Length[ereignis] / Length[omega]]
Clear[k, m, omega, ereignis]
```

```
0.6976
```

7.1.8 Beispiel: Auf $m = 5$ anfangs leere Urnen werden $k = 8$ Kugeln verteilt, wobei jede Kugel mit der gleichen Wahrscheinlichkeit in jede Urne gelangen kann. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei genau $z = 2$ Urnen leer bleiben? (Vergleiche Beispiel 4.2.10.)

▼

Lösung: Das zufällige Verteilen von $k = 8$ Kugeln auf $m = 5$ anfangs leere Urnen besitzt den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_1, x_2, \ldots, x_k \in \{1, 2, \ldots, m\}\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "die erste Kugel wird in die x_1-te Urne gelegt, die zweite Kugel wird in die x_2-te Urne gelegt, …, die k-te Kugel wird in die x_k-te Urne gelegt" entspricht und alle diese Realisierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das Ereignis "es bleiben dabei genau $z = 2$ Urnen leer" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid \text{Mächtigkeit der Menge } \{x_1, x_2, \ldots, x_k\} \text{ ist gleich } m - z\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Length[Union[#]] == $m - z$ & auswählen.

```
k = 8; m = 5; z = 2;
omega = VariationenMitWiederholung[m, k];
ereignis = Select[omega, Length[Union[#]] == m - z &];
N[Length[ereignis] / Length[omega]]
Clear[k, m, z, omega, ereignis]
```

```
0.148378
```

7.1.9 Beispiel: Einer Urne mit $r = 5$ roten und $s = 15$ schwarzen Kugeln werden gleichzeitig $k = 7$ Kugeln entnommen. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei mindestens eine rote Kugel gezogen wird? (Vergleiche Beispiel 4.2.11.)

▼

Lösung: Wir nummerieren die $m = r + s$ Kugeln in der Weise, dass die r roten Kugeln die Nummern $1, 2, \ldots, r$ und die s schwarzen Kugeln die Nummern $r + 1, r + 2, \ldots, m$ erhalten. Unser Zufallsexperiment besteht im gleichzeitigen Ziehen von $k = 7$ Kugeln aus dieser Urne. Dieses Zufallsexperiment besitzt bekanntlich den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_m\} \mid x_i \in \{0, 1\} \text{ und } x_1 + x_2 + \ldots + x_m = k\}$$

wobei die Liste $\{x_1, x_2, \ldots, x_m\} \in \Omega$ der Realisierung "die i-te Kugeln wird genau dann gezogen, wenn $x_i = 1$ ist" entspricht. Alle Realisierungen $\omega \in \Omega$ sind offenbar gleich wahrscheinlich. Das Ereignis "es wird dabei mindestens eine rote Kugel gezogen" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_m\} \in \Omega \mid x_1 + x_2 + \ldots + x_r \geq 1\}$$

Die Menge Ω lässt sich mit dem Befehl KombinationenOhneWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Sum[#[[i]], {i, 1, r}] ≥ 1 & auswählen.

```
r = 5; s = 15; m = r + s; k = 7;
omega = KombinationenOhneWiederholung[m, k];
ereignis = Select[omega, Sum[#[[i]], {i, 1, r}] ≥ 1 &];
N[Length[ereignis] / Length[omega]]
Clear[r, s, m, k, omega, ereignis]
```

```
0.916989
```

7.1.10 Beispiel: Eine Lotterie umfasst $m = 50$ Lose, von denen $g = 5$ gewinnen. Wie groß ist die Wahrscheinlichkeit für (mindestens) einen Gewinn, wenn man $k = 4$ Lose kauft? (Vergleiche Beispiel 4.2.13.)

▼

Lösung: Wir nummerieren die $m = 50$ Lose mit den Zahlen $1, 2, \ldots, 50$, wobei wir den $g = 5$ Gewinnlosen die Nummern 1, 2, 3, 4, 5 zuordnen. Der zufällige Kauf von $k = 4$ Losen lässt sich damit durch den Ereignisraum

$$\Omega = \{\{x_1, x_2, \ldots, x_m\} \mid x_i \in \{0, 1\} \text{ und } x_1 + x_2 + \ldots + x_m = k\}$$

beschreiben, wobei die Liste $\{x_1, x_2, \ldots, x_m\} \in \Omega$ der Realisierung "das i-te Los wird genau dann gekauft, wenn $x_i = 1$ ist" entspricht und alle diese $\omega \in \Omega$ gleich wahrscheinlich sind. Das uns interessierende Ereignis "unter den $k = 4$ gekauften Losen befindet sich mindestens ein Gewinnlos" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_m\} \in \Omega \mid x_1 + x_2 + \ldots + x_g \geq 1\}$$

Die Menge Ω lässt sich mit dem Befehl KombinationenOhneWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Sum[#[[i]], {i, 1, g}] ≥ 1 & auswählen.

```
m = 50; k = 4; g = 5;
omega = KombinationenOhneWiederholung[m, k];
ereignis = Select[omega, Sum[#[[i]], {i, 1, g}] ≥ 1 &];
N[Length[ereignis] / Length[omega]]
Clear[m, k, g, omega, ereignis]
```

```
0.35304
```

7.1.11 Beispiel: Eine Gesellschaft aus 12 Damen und 6 Herren wird in drei Gruppen zu je sechs Personen eingeteilt. Wie groß ist die Wahrscheinlichkeit dafür, dass dabei in jede dieser Gruppen genau zwei Herren gelangen? (Vergleiche Beispiel 4.2.14.)

▼

Lösung: Wir nummerieren die 18 zu vergebenden Plätze mit den Zahlen 1, 2, …, 18, wobei der ersten Gruppe die Plätze 1, 2, …, 6, der zweiten Gruppe die Plätze 7, 8, …, 12 und der dritten Gruppe die restlichen Plätze 13, 14, …, 18 zugeordnet werden. Unser Zufallsexperiment besteht im zufälligen Aufteilen der 6 Herren auf diese 18 Plätze, wofür

$$\Omega = \{\{x_1, x_2, \ldots, x_6\} \mid x_1, x_2, \ldots, x_6 \in \{1, 2, \ldots, 18\} \text{ und } x_1 < x_2 < \ldots < x_6\}$$

ein passender Ereignisraum ist. Die Liste $\{x_1, x_2, \ldots, x_6\} \in \Omega$ entspricht dabei der Realisierung "den 6 Herren werden die Plätze $x_1, x_2, \ldots, x_6$ zugewiesen". Alle diese $\omega \in \Omega$ sind offenbar wieder gleich wahrscheinlich. Das Ereignis "in jeder dieser drei Gruppen befinden sich genau zwei Herren" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_6\} \in \Omega \mid x_1, x_2 \le 6 \wedge 7 \le x_3, x_4 \le 12 \wedge 13 \le x_5, x_6\}$$

Die Menge Ω lässt sich mit dem Befehl KombinationenOhneWiederholungAlt erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung #[[2]] ≤ 6 < #[[3]] ⋀ #[[4]] ≤ 12 < #[[5]] & auswählen.

```
omega = KombinationenOhneWiederholungAlt[18, 6];
ereignis = Select[omega, #[[2]] ≤ 6 < #[[3]] ⋀ #[[4]] ≤ 12 < #[[5]] &];
N[Length[ereignis] / Length[omega]]
Clear[omega, ereignis]
```

```
0.181803
```

7.1.12 Beispiel: $m = 8$ Ehepaare besuchen eine Party. Für ein Tanzspiel werden die Tanzpartner zufällig ausgelost, wodurch jeder Herr jede Dame mit der gleichen Wahrscheinlichkeit zugeteilt bekommt. Wie groß ist die Wahrscheinlichkeit dafür, dass kein Herr mit seiner eigenen Frau tanzt? (Vergleiche Beispiel 4.2.15.)

▼

Lösung: Wir nummerieren die $m = 8$ Ehepaare mit den Zahlen 1, 2, …, 8. Die Menge der möglichen Bildungen von Tanzpartnern lässt sich dann durch die Menge

$$\Omega = \{\{x_1, x_2, \ldots, x_m\} \mid x_i \in \{1, 2, \ldots, m\} \text{ und } x_1, x_2, \ldots, x_m \text{ sind paarweise verschieden}\}$$

beschreiben, wobei die Liste $\{x_1, x_2, \ldots, x_m\} \in \Omega$ der Realisierung "der erste Herr tanzt mit der x_1-ten Dame, der zweite Herr tanzt mit der x_2-ten Dame, …, der m-te Herr tanzt mit der x_m-ten Dame" entspricht und alle diese Realisierungen $\omega \in \Omega$ voraussetzungsgemäß mit der gleichen Wahrscheinlichkeit auftreten. Das uns interessierende Ereignis "kein Herr tanzt dabei mit seiner eigenen Frau" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_m\} \in \Omega \mid x_1 \neq 1, x_2 \neq 2, \ldots, x_m \neq m\}$$

Die Menge Ω lässt sich mit dem Befehl PermutationenOhneWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Apply[And, Table[#[[i]] ≠ i, {i, m}]] & auswählen.

```
m = 8;
omega = PermutationenOhneWiederholung[m];
ereignis = Select[omega, Apply[And, Table[#[[i]] ≠ i, {i, m}]] &];
N[Length[ereignis] / Length[omega]]
Clear[m, omega, ereignis]
```

```
0.367882
```

7.1.13 Beispiel: Ein Teilchen bewegt sich mit der Geschwindigkeit 1 auf der Zahlengerade. Kommt es in einen Punkt mit ganzzahliger Abszisse, so entscheidet sich unser Teilchen (unabhängig von seinem bisherigen Weg) mit gleicher Wahrscheinlichkeit für ein Weiterwandern nach rechts bzw links. Das Teilchen beginnt mit seiner Wanderung im Ursprung. Wie groß ist die Wahrscheinlichkeit dafür, dass sich das Teilchen zum Zeitpunkt $k = 15$ rechts vom Punkt 2 befindet? (Vergleiche Beispiel 4.2.16.)

▼

Lösung: Wir kommen überein, ein Weiterwandern unseres Teilchens nach rechts mit +1 und nach links mit −1 zu bezeichnen. Die Menge der möglichen Wanderungen lässt sich dann durch die Menge

$$\Omega = \{\{x_1, x_2, \ldots, x_k\} \mid x_i \in \{+1, -1\}\}$$

beschreiben, wobei die Liste $\{x_1, x_2, \ldots, x_k\} \in \Omega$ der Realisierung "zum Zeitpunkt $i - 1$ entscheidet sich das Teilchen für ein Weiterwandern nach rechts bzw links je nachdem, ob x_i gleich +1 bzw −1 ist" entspricht und alle diese Realsierungen $\omega \in \Omega$ offenbar gleich wahrscheinlich sind. Das uns interessierende Ereignis "zum Zeitpunkt $k = 15$ befindet sich das Teilchen rechts vom Punkt 2" entspricht somit der Menge

$$A = \{\{x_1, x_2, \ldots, x_k\} \in \Omega \mid x_1 + x_2 + \ldots + x_k > 2\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Apply[Plus, #] > 2 & auswählen.

```
k = 15;
omega = 2 VariationenMitWiederholung[2, k] - 3;
ereignis = Select[omega, Apply[Plus, #] > 2 &];
N[Length[ereignis] / Length[omega]]
Clear[k, omega, ereignis]
```

```
0.303619
```

7.1.14 Beispiel: Mit einer Münze wird so lange geworfen, bis erstmals hintereinander drei Adler auftreten. Wie groß ist die Wahrscheinlichkeit dafür, dass dazu mehr als $w = 15$ Würfe notwendig sind? (Vergleiche Beispiel 4.2.17.)

▼

Lösung: Nach $w = 15$ Würfen lässt sich die gestellte Frage entscheiden. Nun ist aber bekanntlich die Menge

$$\Omega = \{\{x_1, x_2, \ldots, x_w\} \mid x_i \in \{0, 1\}\}$$

ein passender Ereignisraum für das $w = 15$ malige Werfen einer Münze, wobei die Liste $\{x_1, x_2, \ldots, x_w\} \in \Omega$ der Realisierung "beim i-ten Wurf wird genau dann ein Adler geworfen, wenn $x_i = 1$ ist" entspricht und alle diese Realisierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das uns interessierende Ereignis "es sind mehr als $w = 15$ Würfe notwendig, bis erstmals hintereinander drei Adler auftreten" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_w\} \in \Omega \mid x_1 + x_2 + x_3 < 3 \text{ und } x_2 + x_3 + x_4 < 3 \text{ und } \ldots \text{ und } x_{13} + x_{14} + x_{15} < 3\}$$

Die Menge Ω lässt sich mit dem Befehl VariationenMitWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung

Apply[And, Table[#[[i]] + #[[i + 1]] + #[[i + 2]] < 3, {i, w − 2}]] &

auswählen.

```
w = 15;
omega = VariationenMitWiederholung[2, w] - 1;
ereignis = Select[omega, Apply[And, Table[#[[i]] + #[[i + 1]] + #[[i + 2]] < 3, {i, w - 2}]] &];
N[Length[ereignis] / Length[omega]]
Clear[w, omega, ereignis]
```

```
0.323761
```

7.1.15 Beispiel: Vor der Kasse eines Theaters stehen $m = a + b$ Personen, von denen $a = 8$ nur einen 10 € Schein und $b = 12$ nur einen 20 € Schein bei sich haben. Eine Karte kostet 10 €. In der Kasse sind zu Beginn des Kartenverkaufs $u = 5$ 10 € Scheine. Wie groß ist die Wahrscheinlichkeit dafür, dass der Kartenverkauf ohne Stockung abläuft, also in einem Moment, in dem sich in der Kasse kein 10 € Schein befindet, keine Person eine Karte kaufen will, die nur einen 20 € Schein bei sich hat? (Vergleiche Beispiel 4.2.19.)

▼

Lösung: Wir kommen wieder überein, eine Person mit einem 10 € Schein mit +1 und eine Person mit einem 20 € Schein mit −1 zu bezeichnen. Die Menge der möglichen Anordnungen (es kommt nur auf die Art der Scheine an) der m Personen vor der Theaterkasse lässt sich dann durch die Menge

$$\Omega = \{\{x_1, x_2, \ldots, x_m\} \mid x_i \in \{-1, 1\} \text{ und } x_1 + x_2 + \ldots + x_m = a - b\}$$

beschreiben, wobei alle Realsierungen $\omega \in \Omega$ gleich wahrscheinlich sind. Das uns interessierende Ereignis "der Kartenverkauf läuft ohne Stockung ab" entspricht damit der Menge

$$A = \{\{x_1, x_2, \ldots, x_m\} \in \Omega \mid \text{für alle } k \in \{1, 2, \ldots, m\} \text{ gilt } u + x_1 + x_2 + \ldots + x_k \geq 0\}$$

Die Menge Ω lässt sich mit dem Befehl KombinationenOhneWiederholung erzeugen. Die Menge A lässt sich aus dieser Menge Ω mit Select und der Bedingung Min[FoldList[Plus, *u*, #]] ≥ 0 & auswählen.

```
a = 8; b = 12; m = a + b; u = 5;
omega = 2 KombinationenOhneWiederholung[m, a] - 1;
ereignis = Select[omega, Min[FoldList[Plus, u, #]] ≥ 0 &];
N[Length[ereignis] / Length[omega]]
Clear[a, b, m, u, omega, ereignis]
```

```
0.692308
```